Soft Constraint Propagation and Solving in Constraint Handling Rules

S. Bistarelli^{1,3}, T. Frühwirth^2, M. Marte^2, and F. Rossi^3

1 CNR Pisa, Istituto per le Applicazioni Telematiche, (Area della Ricerca di Pisa) Via G. Moruzzi 1, I-56124 Pisa, Italy. Email: bista@di.unipi.it 2 Ludwig-Maximilians-Universität München, Institut für Informatik, Oettingenstr. 67, D-80538 Munich, Germany.

Email: {fruehwir,marte}@informatik.uni-muenchen.de 3 Università di Padova, Dipartimento di Matematica Pura ed Applicata, Via Belzoni 7, I-35131 Padova, Italy.

Email: frossi@math.unipd.it

Abstract. Soft constraints are a generalization of classical constraints, where constraints and/or partial assignments are associated to preference or importance levels, and constraints are combined according to combinators which express the desired optimization criteria. Constraint Handling Rules (CHR) constitute a high-level natural formalism to specify constraint solvers and propagation algorithms. In this paper we present a framework to design and specify soft constraint solvers by using CHR. In this way, we extend the range of applicability of CHR to soft constraints rather than just classical ones, and we provide a straightforward implementation for soft constraint solvers.

Keywords: constraint solving algorithms, soft constraints, constraint handling rules.

1 Introduction

Many real-life problems are easily described via constraints, that state the necessary requirements of the problems. However, usually such requirements are not hard, and could be more faithfully represented as preferences, which should preferably be followed but not necessarily. In real life, we are often confronted with over-constrained problems, which do not have any solution, and this also leads to the use of preferences or in general of soft constraints rather than classical constraints.

Generally speaking, a soft constraint is just a classical constraint plus a way to associate, either to the entire constraint or to each assignment of its variables, a certain element, which is usually interpreted as a level of preference or importance. Such levels are usually ordered, and the order reflects the idea that some levels are better than others. Moreover, one has also to say, via a suitable combination operator, how to obtain the level of preference of a global solution from the preferences in the constraints.

To identify a specific class of soft constraints, one has just to select a certain combination operator and a certain ordered set of levels of preferences. For example, one can choose the set of all reals between 0 and 1, and the min operator (this would be the so-called fuzzy constraints); with this framework, one can give a preference level between 0 and 1 to partial solutions, where a higher level is considered better, and then compute the preference of a global solution as the minimal preference on all constraints. In this view, also classical constraints can be seen as a specific class of soft constraints, where there are only two levels of preference false and true with true better than false), and logical and is the combinator operator.

Many formalisms have been developed to describe one or more classes of soft constraints [6, 7, 3]. In this paper we refer to one which is general enough to describe most of the desired classes. This framework is based on a semiring structure, that is, a set plus two operators: the set contains all the preference levels, one of the operators gives the order over such a set, while the other one is the combination operator [2, 1].

It has been shown that constraint propagation and search techniques, as usually developed for classical constraints, can be extended also to soft constraints, if certain conditions are met [2]. However, while for classical constraints there are formalisms and environments to describe search procedures and propagation schemes [12], as far as we know nothing of this sort is yet available for soft constraints. Such tools would obviously be very useful, since they would provide a flexible environment where to specify and try the execution of different propagation schemes.

In this paper we propose to use the Constraint Handling Rules (CHR) framework [8], which is widely used to specify propagation algorithms for classical constraints, and has shown great generality and flexibility in many application fields. CHR describe propagation algorithms via two kinds of rules, which, given some constraints, either replace them (in a simplification rule) or add some new constraints (in a propagation rule). With a collection of such rules, one can easily specify complex constraint reasoning algorithms.

We describe how to use CHR to specify propagation algorithms for soft constraints. The advantages of using a well-tested formalism, as CHR is, to specify soft constraint propagation algorithms are many-fold. First, we get an easy implementation of new solvers for soft constraints starting from existing solvers for classical constraints. Moreover, we obtain an easy experimentation platform, which is also flexible and adaptable. And finally, we develop a general implementation which can be used for many different classes of soft constraints, and also to combine some of them.

2 Soft Constraints

In the literature there have been many formalizations of the concept of soft constraints [6, 7, 3]. Here we refer to a specific one [2], which however can be shown to generalize and express many of the others. In short, a soft constraint is a constraint where each instantiation of its variables has an associated value from a partially ordered set. Combining constraints will then have to take into account such additional values, and thus the formalism has also to provide suitable operations for combination (×) and comparison (+) of tuples of values and constraints. This is why this formalization is based on the concept of semiring, which is a set plus two operations.

Semirings and SCSPs. A semiring is a tuple ⟨A, +, ×, 0, 1⟩ such that: A is a set and 0, 1 ∈ A; + is commutative, associative and 0 is its unit element; × is associative, distributes over +, 1 is its unit element and 0 is its absorbing element.

In reality, we will need some additional properties, leading to the notion of csemiring (for “constraint-based”): a c-semiring is a semiring ⟨A, +, ×, 0, 1⟩ such that + is idempotent with 1 as its absorbing element and × is commutative.

Let us consider the relation ≤_S over A such that a ≤_S b iff a + b = b. Then it is possible to prove that: ≤_S is a partial order; + and × are monotone on ≤_S ; 0 is its minimum and 1 its maximum; ⟨A, ≤_S ⟩ is a complete lattice and + is its lub. Moreover, if × is idempotent, then: + distributes over ×; ⟨A, ≤_S ⟩ is a complete distributive lattice and × its glb. The ≤_S relation is what we will use to compare tuples and constraints: if a ≤_S b it intuitively means that b is better than a.

In this context, a soft constraint is then a pair ⟨def , con⟩ where con ⊆ V , V is the set of problem variables, and def : D^{|con|} → A. Therefore, a constraint specifies a set of variables (the ones in con), and assigns to each tuple of values of these variables an element of the semiring.

An SCSP constraint problem is a pair ⟨C, con⟩ where con ⊆ V and C is a set of constraints: con is the set of variables of interest for the constraint set C, which however may concern also variables not in con.

Combining and projecting soft constraints. Given two soft constraints c_1 = ⟨def_1, con_1⟩ and c_2 = ⟨def_2, con_2⟩, their combination c_1 ⊗ c_2 is the constraint ⟨def , con⟩ defined by con = con_1 ∪ con_2 and def (t) = def_1(t ↓^{con}_{con1} ) × def (t ↓^{con}_{con2} ), where t ↓^X_Y denotes the tuple of values over the variables in Y , obtained by projecting tuple t from X to Y . In words, combining two soft constraints means building a new soft constraint involving all the variables of the original ones, and which associates to each tuple of domain values for such variables a semiring element which is obtained by multiplying the elements associated by the original soft constraints to the appropriate subtuples.

Given a soft constraint c = ⟨def , con⟩ and a subset I of V , the projection of c over I, written c ⇓_I is the soft constraint ⟨def^′, con^′⟩ where con^′ = con ∩ I and def^′(t^′) = ∑ t/t↓^{con}_{I∩con} =t^′ def (t). Informally, projecting means eliminating some

variables. This is done by associating to each tuple over the remaining variables a semiring element which is the sum of the elements associated by the original constraint to all the extensions of this tuple over the eliminated variables.

Summarizing, combination is performed via the multiplicative operation of the semiring, and projection via the additive operation.

Examples. Classical CSPs are SCSPs where the chosen c-semiring is Bool = ⟨{false, true}, ∨, ∧, false, true⟩. By using this semiring we mean to associate to each tuple a boolean value, with the intention that true is better than false, and we combine constraints via the logical and.

Fuzzy CSPs [6] can instead be modeled by choosing the c-semiring Fuzzy = ⟨[0, 1], max, min, 0, 1⟩. Here each tuple has a value between 0 and 1, where higher values are better. Then, constraints are combined via the min operation and different solutions are compared via the max operation. The ordering here reduces to the usual ordering on reals.

Another interesting instance of the SCSP framework is based on set operations like union and intersection and uses the c-semiring Sets = ⟨℘(A), ∪, ∩, ∅, A⟩, where A is any set. In this case the order reduces to set inclusion and therefore is partial.

It is also important to notice that the Cartesian product of two semirings is again a semiring. This allows to reason with multiple criteria (one for each semiring) at the same time.

Solutions. The solution of an SCSP problem P = ⟨C, con⟩ is the constraint Sol(P ) = ( ⨂ C) ⇓_{con} . In words, we combine all constraints and then we project the resulting constraint onto the variables of interest.

Soft constraint propagation. SCSP problems can be solved by extending and adapting the techniques usually used for classical CSPs. For example, to find the best solution we could employ a branch-and-bound search algorithm (instead of the classical backtracking), and also the successfully used propagation techniques, like arc-consistency, can be generalized to be used for SCSPs.

Instead of deleting tuples, in SCSPs obtaining some form of constraint propagation means changing the semiring values associated to some tuples or domain elements. In particular, the change always brings these values towards the worst value of the semiring, that is, the 0.

The kind of soft constraint propagation we will consider in this paper amounts to combining, at each step, a subset of the soft constraints and then projecting over some of their variables. This is not the most general form of constraint propagation, but it strictly generalizes the usual propagation algorithms like arc- and path-consistency, therefore it is reasonably general.

More precisely, each constraint propagation rule can be uniquely identified by just specifying a subset C of the constraint set, and one particular constraint in C, say c. Then, applying such a rule consists of performing the following operation: c := ( ⨂ C) ↓_{var(c)}. That is, c is replaced by the projection, over its variables, of the combination of all the constraints in C.

It is easy to see that arc-consistency over classical constraints could be modelled by choosing the boolean semiring and selecting C as the set of constraints (two unary and one binary) over any two variables, and c as one of the unary constraints in C.

A soft constraint propagation algorithm operates on a given set of soft constraints, by applying a certain set of constraint propagation rules until stability.

It is possible to prove that any constraint propagation algorithm defined in this way has the following properties [2]:

– it terminates;

– if × is idempotent, then:

• the final constraint set is equivalent to the initial one; • the result does not depend on the order of application of the constraint propagation rules.

Note that if the × operator is not idempotent, like for example in the semiring ⟨R∪+∞, min, +, 0, +∞⟩ for constraint optimizations (where we have to minimize the sum of the costs, and thus × is the sum), we cannot be sure that constraint propagation has the above desirable properties.

3 Constraint Handling Rules (CHR)

CHR (Constraint Handling Rules) [8] are a committed-choice concurrent constraint logic programming language consisting of multi-headed guarded rules that rewrite constraints into simpler ones until they are solved. CHR define both simplification of and propagation over user-defined constraints. Simplification replaces constraints by simpler constraints while preserving logical equivalence. Propagation adds new constraints which are logically redundant but may cause further simplification. CHR have been used in dozens of projects worldwide to implement various constraint solvers, including novel ones such as terminological, spatial and temporal reasoning [8].

In this section we quickly give syntax and semantics for CHR, for details see [8]. We assume some familiarity with (concurrent) constraint (logic) programming [10, 11].

A constraint is a predicate (atomic formula) in first-order logic. We distinguish between built-in (predefined) constraints and CHR (user-defined) constraints. Built-in constraints are those handled by a predefined, given constraint solver. For simplicity, we will regard all (auxiliary) predicates as built-in constraints. CHR constraints are those defined by a CHR program.

Abstract syntax. In the following, upper case letters stand for conjunctions of constraints.

A CHR program is a finite set of CHR. There are two kinds of CHR. A simplification CHR is of the form N @ H <=> G | B and a propagation CHR is of the form N @ H ==> G | B where the rule has an optional name N followed by the symbol @. The multihead H is a conjunction of CHR constraints. The optional guard G followed by the symbol | is a conjunction of built-in constraints. The body B is a conjunction of built-in and CHR constraints.

A simpagation CHR is a combination of the above two kinds of rule, it is of the form

[N ’@’] H1 ’\’ H2 ’==>’ [G ’|’] B.

where the symbol ’\’ separates the head constraints into two nonempty conjunctions H1 and H2. In this paper, a simpagation rule can be regarded as efficient abbreviation of the corresponding simplification rule:

[N ’@’] H1, H2 ’<=>’ [G ’|’] H1, B.

Operational semantics. The operational semantics of CHR programs is given by a state transition system. With derivation steps (transitions, reductions) one can proceed from one state to the next. A derivation is a sequence of derivation steps.

A state (or: goal) is a conjunction of built-in and CHR constraints. An initial state (or: query) is an arbitrary state. In a final state (or: answer) either the built-in constraints are inconsistent or no derivation step is possible anymore.

Let P be a CHR program for the CHR constraints and CT be a constraint theory for the built-in constraints. The transition relation −→ for CHR is as follows.

Simplify H^′ ∧ D −→ (H = H^′) ∧ G ∧ B ∧ D if (H <=> G | B) in P and CT |= D → ∃x̄(H = H^′ ∧ G)

Propagate H^′ ∧ D −→ (H = H^′) ∧ G ∧ B ∧ H^′ ∧ D if (H ==> G | B) in P and CT |= D → ∃x̄(H = H^′ ∧ G)

When we use a rule from the program, we will rename its variables using new symbols, and these variables are denoted by the sequence x̄. A rule with head H and guard G is applicable to CHR constraints H^′ in the context of constraints D, when the condition CT |= D → ∃x̄(H = H^′ ∧ G) holds.

In the condition, the equation (H = H^′) is a notational shorthand for equating the arguments of the CHR constraints that occur in H and H^′. More precisely, by (H_1 ∧. . .∧H_n) = (H^′_1 ∧. . .∧H^′_n) we mean (H_1 = H^′_1)∧. . .∧(H_n = H^′_n), where conjuncts can be permuted. By equating two constraints, c(t_1, . . . , t_n) = c(s_1, . . . , s_n), we mean (t_1 = s_1) ∧ . . . ∧ (t_n = s_n). The symbol = is to be understood as built-in constraint for syntactic equality and is usually implemented by a unification algorithm (as in Prolog).

Operationally, this condition requires to check first whether H^′ matches H according to the definition of the built-in constraints in CT , i.e. whether H^′ is an instance of (more specific than) the pattern H. When matching, we take the context D into account since its built-in constraints may imply that variables in H^′ are equal to specific terms. This means that there is no distinction between, say, c(X) ∧ X = 1 and c(1) ∧ X = 1.

If H^′ matches H, we equate H^′ and H. This corresponds to parameter passing in conventional programming languages, since only variables from the rule head H can be further constrained, and all those variables are new. Finally, using the variable equalities from D and H^′ = H, we check the guard G.

Any of the applicable rules can be applied, but it is a committed choice, it cannot be undone.

If an applicable simplification rule (H <=> G | B) is applied to the CHR constraints H^′, the Simplify transition removes H^′ from the state and adds the body B, the equation H = H^′, and the guard G. If a propagation rule (H ==> G | B) is applied to H^′, the Propagate transition adds B, H = H^′, and G but does not remove H^′. Trivial non-termination is avoided by applying a propagation rule at most once to the same constraints.

4 Implementation

Typically, CHR are used within a CLP environment such as Eclipse or Sicstus Prolog [4]. This means that propagation algorithms are described via CHR, while the underlying CLP language is used to define search procedures and auxiliary predicates. This is the case in our implementation of soft constraints, where the underlying language is Sicstus Prolog. Notice that the actual running code has been slightly edited to abstract away from technicalities like cuts and term copying.

4.1 Choice of the semiring

The implementation is parametric w.r.t. the semiring. To choose one particular semiring S, the user states (that is, asserts) the fact semiring(S) using the predicate use semiring(S), defined as follows:

use_semiring(S):- retract(semiring(_)), assert(semiring(S)).

Semirings S with an idempotent multiplicative operator are recognised by stating the fact idempotent(S). Facts and clauses that are already present to define predicate idempotent are:

idempotent(classical).
idempotent(fuzzy).
idempotent(set).
idempotent((S1, S2)) :- idempotent(S1), idempotent(S2).

The last clause regards semirings obtained by combining two semirings S1 and S2: their cartesian product is idempotent if both components are idempotent.

Recall that a semiring is characterized by ⟨A, +, ×, 0, 1⟩. While the definition of the set A is implicit through the operations, the operations and remaining parameters are defined by CLP clauses.

The two operators of the chosen semiring are defined via predicate plus/3 for the additive operator + and times/3 for the multiplicative operator ×:

plus(W1,W2,W3) :- semiring(S), plus(S,W1,W2,W3).
times(W1,W2,W3) :- semiring(S), times(S,W1,W2,W3).

The partial order of the chosen semiring is defined via predicate leqs/2, in terms of the additive operator, as in the definition of the semiring structure:

leqs(A, B) :- plus(A, B, B).

Finally, the top and bottom element are defined via predicates one/1 and

zero/1.

For example, for the classical semiring (for hard constraints), we have:

plus(classical,W1,W2,W3) :- or(W1,W2,W3).
times(classical,W1,W2,W3) :- and(W1,W2,W3).

   and(t,t,t). and(f,_,f). and(_,f,f).
   or(f,f,f). or(t,_,t). or(_,t,t).

one(t) :- semiring(classical).
zero(f) :- semiring(classical).

Instead, for the fuzzy semiring we have:

plus(fuzzy,W1,W2,W3):- W3 is max(W1,W2).
times(fuzzy,W1,W2,W3):- W3 is min(W1,W2).

one(1) :- semiring(fuzzy).
zero(0) :- semiring(fuzzy).

For the cartesian product of two semirings, we define the operators in terms of the corresponding operators for each semiring:

plus((S1,S2),W1,W2,W3) :-
    W1=(A1,B1), W2=(A2,B2), W3=(A3,B3),
    plus(S1,A1,A2,A3), plus(S2,B1,B2,B3).
times((S1,S2),W1,W2,W3) :-
    W1=(A1,B1), W2=(A2,B2), W3=(A3,B3),
    times(S1,A1,A2,A3), times(S2,B1,B2,B3).

4.2 Domains and constraints

Variable domains are described as lists of pairs, where each pair contains a domain element and an associated preference. The operator in allows to state the unary constraint that a variable is in a certain domain. For example: [X]

in [a-2,b-3].

The operator in can also be used for stating n-ary constraints. For example: [X,Y] in [(a,b)-3,(b,c)-4]. We call such a definition extensional. N-ary constraints can also be defined intensionally, which comes handy in the case of infinite relations. For example, [X,Y] in leq-3-1 associates importance value 3 to all tuples satisfying the constraint leq/2 and value 1 to the others.

Notice that this is just one way to interpret intensional constraints. For leq, we have also experimented with preferences where all pairs that satisfy the relation have maximum preference 1, while for the other pairs the preference could be computed as 1/(1 + d) where d is the difference between the two values that do not satisfy the constraint. Another formula we have used was inspired from work in neural networks: if the importance level of the constraint is l, we give preference level (1 − l)e^{ad} to all the tuples that do not satisfy the constraint, where it is assumed that the preference levels are between 0 and 1 (so 1 − l is the dual of l), and where a is a parameter which allows one to control the steepness of the function.

4.3 Constraint combination

Two extensionally defined soft constraints are combined using the predicate combination/3, which takes two constraints and returns a third constraint which is their combination.

combination(Con1 in Def1, Con2 in Def2, Con3 in Def3):-
        isExtensional(Def1), isExtensional(Def2),
        union(Con1, Con2, Con3),
        semiring(S), zero(Z),
        findall(Con3-W3,
                (member(Con1-W1, Def1),
                 member(Con2-W2, Def2),
                 times(S, W1, W2 ,W3),
                 W3 \== Z),
                Def3).

The combined constraint Con3 in Def3 is computed as follows: the variables involved in the constraint are computed by the union operator. Then findall/3 collects all tuples Con3-W3 of the new constraint in the list Def3, where each tuple is found by computing all pairs of consistent tuples from Con1 and Con2 using member/2 and by computing their preference value W3 using the times operator of the specified semiring S. For performance reasons, the tuples with zero preference value are deleted.

For intensionally defined constraints, a variation of combination/3 is defined, called longcombination/4. It takes an intensionally defined constraint and two extensional domain constraints, and computes a new extensionally defined constraint, which represents the combination of the three original constraints.

longcombination(A in L1, B in L2, E in L4, C in L3):-
    isIntensional(L1), isExtensional(L2), isExtensional(L4),
    union(A, B, AB), union(AB, E, C),
    semiring(S), zero(Z),
    findall(C-W3,
            (member(B-W2, L2),
             member(E-W4, L4),
             checkConstraint(L1, A, W1),
             times(S, W1, W2, W12),
             times(S, W12, W4, W3),
             W3 \== Z),
            L3).

To compute the level of preference that the intensional constraint gives to each tuple, we use the predicate checkconstraint/3, which takes the relation to check (L1), the variables involved (A), the preference parameters of L1, and returns the level of preference for the tuple (W1). For example:

checkConstraint(leq-W-WA, [X,Y], W1):- X=<Y -> W1 is W ; W1 is WA.
checkConstraint(slq-W-WA, [X,Y], W1):- X=<Y -> W1 is W ; W1 is
                                               max(WA,1/(X-Y)*W1).

The first relation assigns weight W to each tuple that satisfies the relation X=<Y, and W1 to the other tuples. The second relation assigns to each tuple a weight which depends on the distance between X and Y.

4.4 Constraint projection

Predicate projection/3 implements the projection operator for an extensionally defined soft constraint and a list of variables Con2, resulting in a new constraint

Con2 in Def2.

projection(Con1 in Def1, Con2, Con2 in Def2):-
        isExtensional(Def1),
        findall(Con2-W1, (member(Con1-W1, Def1)), Def3),
        keysort(Def3, Def4),
        semiring(S),
        allplus(Def4, Def2, S).

First findall/3 finds all tuples in terms of the variables of interest Con2 using the tuples from the original constraint Con1 in Def1. These tuples are sorted so that tuples with the same domain element are neighbouring. Then predicate allplus/3 sums all the semiring values whose domain element is the same to compute the final new domain Def2.

allplus([], [], _).
allplus([A-W1, A-W2|Def0], Def, S) :-
        !, plus(S, W1, W2, W3), allplus([A-W3|Def0], Def, S).
allplus([A-W1|Def0], [A-W1|Def], S):- allplus(Def0, Def, S).

4.5 Node- and arc-consistency

A variable is node-consistent if for every value in the current domain of the variable, each unary constraint on the variable is satisfied. The following CHR rule achieves node-consistency by intersecting the domains associated with the variable X using combination/3:

node_consistency @ Con in Def1, Con in Def2 <=>
    Con=[X],
    isExtensional(Def1), isExtensional(Def2) |
    combination(Con in Def1, Con in Def2, Con in Def3),
    Con in Def3.

Actually, we can drop Con=[X] from the guard of the rule, so that Con can be any list of variables. Thus the generalized rule now performs intersection of the domains of two soft constraints over the same list of variables.

The following simpagation rule implements arc-consistency, by combining binary and unary constraints involving two variables X and Y and then projecting onto each of the two variables. In effect, the two unary constraints on X and Y are tightened taking into account the binary constraint.

arc_consistency @ [X,Y] in C \ [X] in A, [Y] in B <=>
    isExtensional(C), isExtensional(A), isExtensional(B),
    semiring(S), idempotent(S) |
      combination([X,Y] in C, [X] in A, [X,Y] in D),
      combination([X,Y] in D, [Y] in B, [X,Y] in E),
      projection([X,Y] in E, [X], [X] in F),
      projection([X,Y] in E, [Y], [Y] in G),
    [X] in F, [Y] in G.

We recall here that soft arc-consistency can be applied only when the (multiplicative operation of the) semiring is idempotent. Otherwise, in our implementation, we apply a variation of arc-consistency that uses another projection predicate. It eliminates from the domains only those elements with zero as associated preference level.

Another version of the arc-consistency rule dealing with intensionally defined constraint has also been implemented. It basically differs from the rule above only

in that it uses the goal longcombination([X,Y] in C, [X] in A, [Y] in B,

[X,Y] in E) instead of the two goals involving combination/3.

4.6 Complete solvers

Naive solver. Predicate solve/2 implements the notion of solution, by combining all the constraints in Cs and then projecting over the variables of interest (those in Con).

solve(Cs, Con, Solution) :-
    globalCombination(Cs, C), projection(C, Con, Solution).

Predicate globalCombination/2 folds combination/3 over a list of constraints.

A variant of solve with two arguments computes the list of constraints Cs by looking in the current constraint store.

Solver based on dynamic programming. This solver, called dp, incrementally eliminates a set of variables from the constraint store. It is working on one variable at a time. First, it selects a variable X to eliminate (but not one from those given in the list Xs). Second, it identifies the constraints involving X and combines them into a single constraint Cs. Third, it eliminates X from Cs by projection obtaining C. Finally, the constraints involving X are replaced by C and the solver iterates to eliminate the remaining variables.

dp(Xs) :-
   (selectVariable(X,Xs),
   ->
    findall_constraints(X, _ in _, Cs),
    globalCombination(Cs, C0),
     C0 = (Con0 in _),
     delete(X, Con0, Con1),
    projection(C0, Con1, C),
    removeConstraints(Cs0),
    addConstraints(C),
    dp(Xs)
   ;
    true
   ).

Solver based on branch & bound with variable labeling. This solver, called VARBB, performs branch and bound with variable labeling in the search for a solution with maximal weight. Given a list of variables Xs and constraints Con, the solution Solution is found in the following way: first a variable X is selected deterministically from Xs according to some built-in strategy. Second, a value-weight pair is chosen non-deterministically from the domain of X according to some built-in strategy. Then the resulting unary constraint [X] in [A-AW] is imposed. If there is already a current bound (weight), the constraints Con are solved using solve and it is made sure that there is at least one possible value in the solution domain whose weight is lower than the current weight. Finally, the recursive call continues with the remainder of the variables Xs1.

If the list of variables is empty, the second clause for varbb computes a solution and updates the bound to be the weight occurring in the solution.

varbb(Xs, Con, Solution) :-
        selectVariable(Xs, X, Xs1),
        selectValue(X, A-AW),
        [X] in [A-AW],
        ( bound(LB)
        -> solve(Con, _ in Def),
            once((member(_-W, Def), less(LB, W)))
        ; true),
        varbb(Xs1, Con, Solution).
varbb([], Con, Solution) :-
        solve(Con, Solution),
        Solution = (_ in [_-B]),
        update(bound(B)).

5 Examples

5.1 Domains, constraints, propagation, solvers

The following goal defines some variable domains and constraints, and then it calls the naive solver with X as the variable of interest:

[X] in [[a]-2,[b]-3,[c]-4], [Y] in [[a]-1,[c]-5],
[X,Z] in [[a,b]-3, [a,c]-1, [b,a]-3], solve([X], S).

The following two goals trigger constraint combination over two constraints:

combination([X,Y] in [(a,b)-3, (a,c)-1, (b,a)-3],
X in [a-2,b-3,c-4], C).
combination(X in [a-2,b-3,c-4], Z in [a-1,c-5], (X,Z) in L).

Both variable domains and constraints can be intensional and extensional. Here is an example with an intensional binary constraint and two extensional domains:

(X,Z) in leq-2, X in [1-2,2-3,3-4], Z in [1-1,3-5].

Constraints can be intensional, as in

(Y,X) in leq-2, X in [1-1,2-1,3-1], Y in [2-1,3-1,4-1].

where variables X and Y are linked by constraint leq with importance level 2.

When using the cartesian product of two semirings, preference values are pairs instead of single elements, as in

use semiring((fuzzy,fuzzy)), X in [a-(2,3),b-(3,4),c-(4,6)],
X in [a-(1,9),c-(5,5)].

where we use the Cartesian product of two fuzzy semirings.

Projection can be performed also over more than one variable:

projection((X,Y,Z) in [(a,b,c)-2], [X,Y], [X,Y] in L2).

Combination can generate also non-binary constraints:

combination([X,Y] in [(a,b)-3], (Y,Z) in [(b,c)-4], (X,Y,Z) in L3).

Here is an example of the use of the set semiring:

semiring(set), X in [a-[a],b-[a,b],c-[a,b,c]],
X in [a-[a,b],c-[a,b,d]].

5.2 Robot dressing

This problem concerns a robot that has to choose its clothes: he can choose one shirt (F), one pair of trousers (T) and one pair of shoes (S). The original version of the problem has been developed by Z. Ruttkay for her fuzzy solver, then it has been adapted by Y. Georget for his soft CLP language clp(fd,S) over the fuzzy semiring [9].

robot([F,T,S]) :-
  use_semiring(fuzzy),
  F in [0-10,1-10],
  S in [0-10,1-10],
  T in [0-10,1-10,2-10],
  (F,T) in [(0,0)-10,(0,1)-4,(0,2)-2, (1,2)-8,(1,1)-5],
  (F,S) in [(0,1)-10,(0,0)-7,(1,0)-10,(1,1)-1],
  (T,S) in [(0,0)-10,(0,1)-7,(1,0)-10,(1,1)-4,(2,1)-10,(2,0)-6].

The constraints are all defined extensively: each pair of values of the variables are associated to a preference value. Preference values are between 1 and 10, internally normalized between 0 and 1 to use the fuzzy semiring.

5.3 Different solvers

We now show the input-output behaviour for one particular constraint problem on which we use different solvers: just soft constraint propagation using nodeand arc-consistency, B&B with constraint labeling, B&B with variable labeling, and dynamic programming:

– soft constraint propagation:

?- Def = [[1]-0.1, [2]-0.2, [3] -0.3, [4]-0.1]),
   [X] in Def,
   [Y] in Def,
   [X,Y] in [[1,1]-0.1, [2,1]-0.2, [3,2] -0.3, [4,4]-0.1].

[X,Y] in [[1,1]-0.1,[2,1]-0.2,[3,2]-0.3,[4,4]-0.1],
[X] in [[1]-0.1,[2]-0.1,[3]-0.2,[4]-0.1],
[Y] in [[1]-0.1,[2]-0.2,[4]-0.1].

– branch & bound with constraint labeling (same Def as above):

?- Def = ..., [X] in Def, [Y] in Def,
   [X,Y] in [[1,1]-0.1, [2,1]-0.2, [3,2] -0.3, [4,4]-0.1],
   conbb(S).

S = [Y,X] in [[1,1]-0.1],
[Y] in [[1]-0.1],
[X] in [[1]-0.1],
[X,Y] in [[1,1]-0.1]
;
S = [Y,X] in [[2,3]-0.2],
[X] in [[3]-0.2],
[Y] in [[2]-0.2],
[X,Y] in [[3,2]-0.3].

– branch & bound with variable labeling (same Def and same binary constraint):

?- Def = ..., [X,Y] in ...,
   varbb([X, Y], [X, Y], S).

S = [X,Y] in [[1,1]-0.1],
[X,Y] in [[1,1]-0.1,[2,1]-0.2,[3,2]-0.3,[4,4]-0.1],
[X] in [[1]-0.1],
[Y] in [[1]-0.1]
;
S = [X,Y] in [[3,2]-0.2],
[X,Y] in [[1,1]-0.1,[2,1]-0.2,[3,2]-0.3,[4,4]-0.1],
[X] in [[3]-0.2],
[Y] in [[2]-0.2].

– dynamic programming (same Def and same binary constraint):

?- Def = ..., [X,Y] in ...,
   dp([X]).

[X] in [[1]-0.1,[2]-0.1,[3]-0.2,[4]-0.1].

6 Conclusions

We have implemented a generic soft constraint environment where it is possible to work with any class of soft constraints, if they can be cast within the semiring-based framework: once the semiring features have been stated via suitable clauses, the various solvers we have developed in CHR and Sicstus Prolog will take care of solving such soft constraints. We have implemented semi-rings for classical, fuzzy, set, and Cartesian-product soft constraints. Our solvers include propagation-based node- and arc-consistency solvers as well as the several complete solvers using branch and bound with variable or constraint labeling or dynamic programming. The solvers will soon be available online at (http://www.pms.informatik.uni-muenchen.de/~webchr/).

We plan to predefine more classes of soft constraints and to develop other soft propagation algorithms and solvers for soft constraints.

We also plan to compare our approach to the one followed by the soft constraint programming language clp(fd,S) [9]. Of course we do not expect to show the same efficiency as clp(fd,S), but we claim the same generality, and a very natural envornment to develop new propagation algorithms and solvers for soft constraints. Moreover, we did not need to add anything, except the clauses and CHR rules shown in this paper, w.r.t. the existing CHR environment and CLP language of choice. On the other hand, clp(fd,S) needed a new implementation and abstract machine w.r.t. clp(fd) [5], from which it originated.

References

1. S. Bistarelli. Soft Constraint Solving and programming: a general framework. PhD thesis, Dipartimento di Informatica, University of Pisa, 2001.

2. S. Bistarelli, U. Montanari and F. Rossi. Semiring-based Constraint Solving and Optimization. Journal of ACM, vol. 44, no. 2, March 1997.

3. A. Borning, M. Maher, A. Martindale, and M. Wilson. Constraint hierarchies and logic programming. In Martelli M. Levi G., editor, Proc. 6th International Conference on Logic Programming, pages 149–164. MIT Press, 1989.

4. M. Carlsson and J. Widen. SICStus Prolog User’s Manual. On-line version at http://www.sics.se/sicstus/. Swedish Institute of Computer Science (SICS), 1999.

5. P. Codognet and D. Diaz. Compiling constraints in CLP(FD). Journal of Logic Programming, vol. 27, n. 3, 1996.

6. D. Dubois, H. Fargier and H. Prade. The calculus of fuzzy restrictions as a basis for flexible constraint satisfaction. Proc. IEEE International Conference on Fuzzy Systems, IEEE, pp. 1131–1136, 1993.

7. E. C. Freuder and R. J. Wallace. Partial constraint satisfaction. AI Journal, 58, 1992.

8. T. Frühwirth, Theory and Practice of Constraint Handling Rules, Special Issue on Constraint Logic Programming (P. J. Stuckey and K. Marriot, Eds.), Journal of Logic Programming, Vol 37(1-3):95-138 Oct-Dec 98.

9. Y. Georget and P. Codognet. Compiling Semiring-based Constraints with clp(FD,S). Proceedings of CP’98, Springer, 1998.

10. K. Marriott and P. J. Stuckey. Programming with constraints: an introduction.

MIT Press, 1998.

11. P. Van Hentenryck. Constraint Satisfaction in Logic Programming. MIT Press, 1989.

12. P. Van Hentenryck and al. Search and Strategies in OPL. ACM Transactions on Computational Logic, 1(2), October 2000.